Acta Materialia Sorbonica **1**, 1, 1-10 (2013)

dated : December 10, 2012
revised : January 4, 2013

Keywords:
Planetary Ball Milling
milling balls movement,
relative velocities, shocks.

Planetary Ball Milling: Geometry-Based Considerations

Dominique Vrel
Université Paris 13, Sorbonne Paris Cité, Laboratoire des Sciences des Procédés et des Matériaux, CNRS - UPR 3407, 99 Avenue J.-B. Clément, 93430 Villetaneuse, France

Abstract: Using simple calculations and a spreadsheet based implementation, velocities and movement of the balls present in a jar are computed. From these calculations, considerations regarding the wear and tear of the milling media, efficiency of the milling and the nature of the shocks, i.e. frontal shocks or not, are deduced.

I. INTRODUCTION

Planetary Ball Milling is now a well known process to produce fine-grained materials, down to the nanometer range. However, even if theoretical considerations exist, depicting the ideal conditions for milling as far as balls diameter, rotation velocities, ball-to-powder ratio, few papers explicitly describe the theoretical movement of the balls during milling, and the impact this has on shock intensities and frequency. Using here simple geometrical considerations, the general movement of the balls during milling is described.

II. PLANETARY MOVEMENT

Planetary ball milling takes its name from the jars' movement during milling, which mimics the movement of the planets around the sun, or of a natural satellite around its planet. Indeed, the jar is set on a plate, called for that reason the "solar plate", and has a general circular movement around the center of that plate at an angular velocity Ω. At the same time, the jar rotates around itself (but with no angle from the ecliptic plane) at another angular velocity ω. Most planetary ball mills have a fixed $\frac{\omega}{\Omega}$ ratio, set at -2, and the general movement of the jars is described on Figure 1.

However, one has to be careful when discussing this ratio: the minus sign simply depicts that the jar rotates around itself in the opposite direction than its general movement around the "sun", and the 2 value indicates that in one rotation around this sun, each point of the jar would have faced the sun twice; to continue with this sun-planet comparison, one could say that the jar has a "year" of two "days".

But when looking carefully at the figure, one would realize that even though there is two days in one year, the jar actually makes only one rotation around itself *if you consider the whole movement in a Cartesian coordinate system.* Therefore, as all calculations will be performed in such a coordinate system, we have to correct the ω angular velocity by putting

$$\omega' = \omega + \Omega, \tag{1}$$

as ω and Ω have opposite signs.

Let's note immediately that positive $\frac{\omega}{\Omega}$ ratios will have little meaning, as calculations would prove hereunder: for a value of 1, the movement of the jar would be similar as the one of the moon around Earth, showing always the same face to its sun, and all the balls would be stuck by centrifuge forces in the outer part of the jar.



FIG. 1. Jar movement in a "classical" ball-milling device, using an angular velocities ratio $\frac{\omega}{\Omega} = -2$.The center of the jar has a general circular movement around its "'sun" at an angular velocity Ω, depicted by the thin blue line; it also rotates around itself with an angular velocity ω. The resulting movement of a point at the surface of the jar is then depicted by the thin red line.

III. CALCULATION HYPOTHESES

The basics of all the results presented hereunder are very simple and are based on three assumptions:

- the ball mill is supposed to rotate fast enough for the ball movement to depend mainly of the jar movement, therefore gravity is neglected, and the balls' movement is therefore considered to be 2-dimensional;
- if the ball is stuck at the inner surface of the jar, the velocity of the ball is calculated from the evolution of the jar position;
- if the ball is not stuck at the inner surface of the jar, its movement is free and its movement in the Cartesian coordinate system is linear, at a constant velocity .

From these assumptions, it becomes quite clear that we need a criterion telling us if the ball has a free movement or if it is stuck to the inner wall of the jar. This is done through the following procedure:

- let's assume that the ball is free; therefore, its velocity is constant and equal to the one it had at the previous time step;
- calculate from that velocity the new position of the ball;
- calculate from the circular movement of the jar, the new position of the center of the jar;
- calculate the distance from the center of the jar to the ball position;
- if this distance is greater than the inner radius of the jar *minus* the radius of the ball, then the movement cannot be free, and the ball is (still) stuck to the inner wall of the jar, its movement is controlled by the double rotation of the jar (i.e. no slipping). Otherwise, the ball is really free, and its new position is calculated from the previously calculated velocity, which is kept constant for the next time step.

IV. IMPLEMENTATION

Using Microsoft Excel® software, here is how the calculation is performed:

A. Starting conditions

In the A1-B1 cells, we enter the time step:

```
A1> Deltat
B1> 0.01
```

In the A2-B2, we enter the value of Ω, and in the A3-B3 the value of of ω; for calculation purposes, we immediately calculate the corrected value ω' in the A4-B4 lines

```
A2> Omega
B2> 1
A3> omega
B3> -2
A4> omega (corrected)
B4> =B3+B2
```

We currently use the Fritsch® *Pulverisette 7 Premium line* planetary ball mill; in that mill, the radius from the center of the solar plate to the center of the jar is 7.5cm, whereas the inner radius of the jar is 2.5cm. Balls with diameters of 0.3, 0.5, 1, 1.5 and 2cm could be used. Once more, the distance from the center of the jar to the center of the ball is therefore limited to the inner jar radius *minus* half the ball diameter. Let's compute this:

```
D1> R
E1> 7.5
G1> Ball diameter (.3, .5, 1, 1.5 or 2)
H1> 0.3
G2> internal radius of the jar
H2> 2.5
D2> r
E2> =H2-H1/2
```

B. Evolution with time

In the first column, we compute the time, and the position of Θ, resulting from the angular velocity Ω:

```
 A8> Time
 A9> 0
A10> =A9+$B$1
 B8> Theta
 B9> =A9*$B$2
B10> =B9+$B$1*$B$2
```

The cells A10 and B10 are then copied in a great number of cells underneath (e.g. 4000) using the CTRL B command.

For the θ value, resulting from the ω rotation of the jar around itself, things become a little tricky, for we have to consider if the ball is free or not. We just assume it is not as an initial condition, and write the following cells:

```
 C8> theta
 C9> =A9*$B$4
C10> =C9+$B$4*$B$1
```

But for the following lines, we need to set up a flag indicating whether the ball is free or not, setting an arbitrary value of 0 (the ball is not free) for the first time steps; this flag will be equal to 1 if the ball is free. Because of how this flag will be calculated, we put it in column I:

```
I8> Free ?
I9> 0
```

For the initial ball and jar position, we set an arbitrary position at the west of the diagram, and we can calculate the position of the supposedly free ball at the next time step:

```
 D8> x ball (if free)
 E8> y ball (if free)
 D9> =-E1-E2
 E9> 0
 F8> x jar
 G8> y jar
 F9> =-$E$1*COS(B9)
 G9> =$E$1*SIN(B9)
D10> =D9
E10> =SIN(B10)*$E$1-SIN(C10)*$E$2
```

As the jar position does not need any other piece of information, we already can copy cells F9-G9 until the last line of our table.

From now on, you must be aware that I am using a French version of Excel; therefore, if you try to implement the code provided here on another version some translation has to be performed: RACINE is a function providing the square root (it probably is translated by ROOT in English), SI is a conditional function (IF ?), and I believe some semicolons should be replaced by commas.

Let's calculate in column H the distance between the center of the jar and the center of the ball, assuming this ball is free:

```
H8> Distance between the jar and the
    free ball
H9> =RACINE((F9-D9)^2+(G9-E9)^2)
```

Although we have not calculated the values of θ or the coordinate of the free ball for time steps other than the two first, we know that the calculation of the H column will always be done the same way, and we can copy the H9 cell in the lines under it.

We now have a criterion to determine whether the ball is free or not, we enter it in the I10 cell and copy it in all the lines under it:

```
I10> =SI(H10>$E$2;0;1)
```

To calculate the position of the ball, whether it is free or not, we are going to cheat a little by using the flag calculated in column I:

```
 K8> x ball
 L8> y ball
 K9> =D9
 L9> =E9
K10> =I10*(D10)+(1-I10)*(F10-$E$2*COS(C
     10))
L10> =I10*E10+(1-I10)*(G10+$E$2*SIN(C10
     ))
```

Indeed, the coordinates of the ball is equal to (Dxx,Exx) if the flag Ixx is set to 1, and equal to the coordinates of the center of the jar to which the rotation of the θ angle has to be added, if the flag is set to 0. Cells K10-L10 are then copied in all the lines under them.

Finally, using the known positions of the ball, at the current time step and the previous one, we can calculate the velocity of the ball in the Cartesian coordinate system, except for the very first time step:

```
 M8> x' ball
 N8> y' ball
M10> =K10-K9
N10> =L10-L9
```

Once more, cells M10 and N10 have to be copied underneath. You may have noted that technically, to really compute the velocity of the balls, we should divide the values entered in the M10 and N10 cells by the time step. However, as this time step is constant, and as to calculate the new position of the ball we would have to re-multiply by the time step, we just ignore both calculations for the sake of simplicity, but keep in mind that these values are not to be considered "as is".

The main part of the spreadsheet is now complete, except for three columns, namely columns C, D and E, for which we only entered the two first time steps.

For columns D and E, things are now quite simple, as they should contain the ball position if it is free, i.e. if it has a constant velocity.

```
D11> =K10+M10
E11> =L10+N10
```

Note that for accuracy reasons, we add the ball velocity not to the (D10,E10) coordinates, as they are valid only if the ball was already free in the previous time step. And once more, the D11 and E11 cells are copied underneath.

The only missing data now concern the value of θ, for which some specific attention should be taken:

- if the ball is not free, then it will be stuck at the inner surface of the jar, and θ will simply increase as a function of the ω value and the time step:

```
C11> =C10+$B$4*$B$1
```

- But if the ball is free, the movement of the ball is linear in the Cartesian coordinate system and does not depend on θ; on the contrary, it is θ that should be calculated as a function of the jar center position and the ball position:

$$\theta = -tan^{-1}\left(\frac{y_{ball} - y_{jar}}{x_{ball} - x_{jar}}\right)$$

if $x_{ball} > x_{jar}$, and

$$\theta = \pi - tan^{-1}\left(\frac{y_{ball} - y_{jar}}{x_{ball} - x_{jar}}\right)$$

if $x_{ball} < x_{jar}$.

This second condition could be implemented as :

```
C11> SI(K10<F10;0;PI())-ATAN((L10-G10)/
     (K10-F10))
```

Combining the two above mentioned possibilities for the C11 cell yields to:

```
C11> =SI(I10=0;C10+$B$4*$B$1;SI(K10<F10
     ;0;PI())-ATAN((L10-G10)/(K10-F10)))
```

and, as always, the C11 cell is copied downwards.

V. RESULTS

A. Movement of a single ball

For this section, we will consider the relative movement of a single ball, having a diameter of 0.3cm, i.e. the smallest available commercial ball diameter.



FIG. 2. Movement of the ball during the first full rotation of the jar (in red). The movement of the jar is depicted in blue. Axis represent the coordinates in cm.

1. *General movement*

Using a δt value of 2π divided by the number of line, one would get, representing the L columns as a function of the K column, a representation of the ball movement in a Cartesian coordinate system when the jar makes it first rotation around the solar center. Such a movement is here represented in Figure 2.

One can clearly observe on this figure that the ball has a linear movement before being submitted to a first shock, then is again accelerated along a curved path before reaching again a linear movement, etc. During the first rotation of the jar, the ball is submitted to three shocks.

To gain a better overview of the ball movement, Figure 3 depicts the movement of the ball during the 10 first rotations. From that figure, it also becomes obvious that the ball hits the jar not in its most outside point (e.g. the ball never comes close to the ordinate y=9.85, whereas it was at x=-9.85 as an initial condition), and therefore that the initial condition we chose is not quite representative for a steady-state movement. When making other calculations concerning the ball movement in the following paragraphs, we will preferentially consider the ball movement between two consecutive shocks, neglecting the part of the movement between the start of the calculation and the first shock.





FIG. 3. Movement of the ball during the first 10 full rotations of the jar (in red). The movement of the jar is depicted in blue. Axis represent the coordinates in cm.

FIG. 4. Modulus of the ball velocity as a function of the Θ angle during the first rotation. Indicated values have to be multiplied by the RPM value to obtain velocity in $m \cdot s^{-1}$.

2. *Maximum velocity*

To calculate the velocity of the ball, we add another column by copying the O10 cell downwards:

```
 O8> Ball velocity
O10> =RACINE(M10^2+N10^2)/$B$1
```

The result of such a calculation would be a velocity in $cm \cdot s^{-1}$, provided that the Ω value is set in $rad \cdot s^{-1}$ and the diameter in cm. Instead, most commercial apparatus display an Ω value in RPM (rotation per minute), and the velocity should rather be displayed in $m \cdot s^{-1}$. To convert the above calculation in the proper units, this result has to be divided by 100 (to convert cm in m), multiplied by 2π (to convert RPM in $rad \cdot min^{-1}$) and divided by 60 (to convert $rad \cdot min^{-1}$ in $rad \cdot s^{-1}$).

The resulting calculation is thus as follow:

```
O10> =RACINE(M10^2+N10^2)/$B$1/3000
     *PI()
```

Note that the calculation has here been made for an Ω value of 1 RPM (and an ω value of -2RPM); to adapt the results in reasonable milling conditions, this velocity has to be multiplied by the real RPM value. Indeed, this is the main interest in planetary ball milling, as opposed to regular circular vertical milling: the ball movement is independent from the chosen velocity, and is not limited by centrifugal forces. As an example, choosing an Ω value of 350RPM would yield a maximum velocity of $3.126 m \cdot s^{-1}$. A representation of the ball velocity modulus, still for an Ω value of 1RPM, is represented in Figure 4.

3. *Relative ball velocity at impact*

In the above calculation of the velocity, one would have noticed that the velocity of the ball is computed in the Cartesian coordinate system, but, as the jar is also in movement, the energy impact between the ball and the jar cannot be straightly calculated.

For that reason, we will now compute the position of the ball *in the rotating jar referential*. To do such thing, we will introduce two more columns:

```
Q9> =RACINE((K9-F9)^2+(L9-G9)^2)*
    COS(C9-B9)
R9> =RACINE((K9-F9)^2+(L9-G9)^2)*
    SIN(C9-B9)
```

The resulting movement of the ball is then represented in Figure 5.

From this figure, we see that the ball is in about half of its movement displaced at the inner surface of the jar. When it becomes free, it keeps its linear movement *in a Cartesian coordinate system*, but in the present referential, rotating with the Θ angle, it behaves like it is moving backwards to hit the jar almost perpendicularly.

As for the ball relative velocity to the jar, it may be calculated in column S as follow:

```
S10> =RACINE((Q10-Q9)^2+(R10-R9)^2)
```

However, we should apply the same treatment as before to the velocity, in order to reach more easily velocities in $m \cdot s^{-1}$:

```
S10> =RACINE((Q10-Q9)^2+(R10-R9)^2)/
```



FIG. 5. Movement of the ball in a coordinate system rotating with the jar.

```
$B$1/3000*PI()
```

Figure 6 represents this velocity as a function of the Ω angle for the first full rotation of the jar around the solar center.



FIG. 6. Relative velocity of the ball to the jar, as a function of Θ.

Contrarily to what we had obtained with figure 4, here the velocity is constant when the ball is stuck at the jar inner surface, as it simply correspond to the rotation velocity of the jar around itself, ω. When the ball becomes free, as seen on figure 5, the ball comes closer to the center of the jar, and the relative velocity reaches a minimum value. As the relative movement of the ball from the jar progressively comes close to 90°, the relative velocity increases to reach a maximum value at impact. Once more, the represented values should be multiplied by the RPM values to yield velocities in $m \cdot s^{-1}$.

4. Impact angle

To determine the impact angle of the ball on the inner jar surface, the scalar product of the velocity of the ball to the radius of the jar is calculated, and, from that product, the angle of the jar movement is deduced. Figure 7 represents this angle as a function of Ω during the first jar rotation.



FIG. 7. Angle (in °) of the ball movement to the jar radius, as a function of Θ.

This figure clearly displays a minimum angle of 90°, i.e. when the ball is stuck to the jar inner wall. When free movement starts, the angle steadily increases, until the shock happens, for a value very close to 360°, i.e. the ball movement is almost parallel to the jar diameter, i.e. the ball hits the jar almost perpendicularly to its inner surface.

5. Impact frequency

Impact frequency may then be calculated from the time separating two consecutive shocks. As mentioned

Ball diameter (cm)	Maximum velocity $(m \cdot s^{-1} \cdot RPM^{-1})$	Max. relative velocity to the jar $(m \cdot s^{-1} \cdot RPM^{-1})$	Impact angle (°)	shock “frequency” (RPM^{-1})	Shock “energy” (see text)
0.3	$8.93 \cdot 10^{-3}$	$8.084 \cdot 10^{-3}$	359.68 = -0.31	3.302	$5.826 \cdot 10^{-6}$
0.5	$8.85 \cdot 10^{-3}$	$7.761 \cdot 10^{-3}$	356.73 = -3.26	3.406	$2.864 \cdot 10^{-5}$
1.0	$8.65 \cdot 10^{-3}$	$6.916 \cdot 10^{-3}$	348.64 = -11.46	3.736	$1.786 \cdot 10^{-4}$
1.5	$8.47 \cdot 10^{-3}$	$6.071 \cdot 10^{-3}$	340.02 = -19.98	4.161	$5.176 \cdot 10^{-4}$
2.0	$8.31 \cdot 10^{-3}$	$5.217 \cdot 10^{-3}$	331.07 = -28.92	4.744	$1.032 \cdot 10^{-3}$
powder	$9.07 \cdot 10^{-3}$	$8.583 \cdot 10^{-3}$	6.24	3.156	NA

TABLE I. Milling parameters with different ball diameters.



FIG. 8. Ball trajectory during the first full revolution of the jar in the Cartesian coordinate system. In blue is represented the trajectory of the center of the jar; in red, from bright to dark, are represented the trajectory of a ball with a 0.3, 0.5, 1.0, 1.5 and 2.0 cm.



FIG. 9. Ball trajectory during the first full revolution of the jar in the coordinate system associated with the rotating jar. In red, from bright to dark, are represented the trajectory of a ball with a 0.3, 0.5, 1.0, 1.5 and 2.0 cm.

earlier, the time before the first shock must be ignored, as the initial conditions are not representative of a position of the ball steady-state movement. For the currently used ball diameter (0.3cm), the shock frequency would be 3.406 shock/RPM.

B. Influence of the ball diameter

1. *Using commercially available milling balls*

Commercially available milling balls are available in the following dimensions: 0.3, 0.5, 1.0, 1.5 and 2cm. In order to add these possible experimental conditions, the H2 cell is modified accordingly. Results of the different values obtained by the simulations are reported in Table I.

In order to get an idea of the variation of the shock energy per ball and per time unit, we added a last column in table I, representing the square of the maximal relative velocity, multiplied by the shock frequency, and multiplied by the ball volume. Note that for this result to be in readable units, one should take into account the number of balls present in the jar (naturally, you may fit more small balls that big ones) and also the ball density. However, this data provides relative variations when changing the ball diameter.

Another point that should be raised, is that the trajectory of the balls change significantly with their diameter. Figures 8 and 9 represent the ball trajectories according to their diameter, on which we can see the variation of the impact frequency and of the impact angle.

From these figures, one should point out that it is clearly not advised to mix different ball diameter during a milling: as their trajectory differ significantly from each other, mixing different sizes would induce ball-to-ball interaction, including shocks during their free trajectory. From these inter-ball shocks, the efficiency of the milling would be decreased, with a strong variation of the impact angle at the inner surface of the jar, and significant abrasion of the balls would induce a greater contamination of the milled powder by the grinding ball material. On the contrary, if only one ball diameter is used, purely geometrical-based calculations suggest that the grinding balls simply follow each other; ball-to-ball interaction may thus be neglected, and the grinding efficiency only comes from ball-to-jar interactions.

2. *Considerations on the powders*

When considering the milled powders, one may assume that their movement could also be controlled according to the same laws; in such a case, the movement of the powders may be represented considering a ball-equivalent diameter ranging from 10^{-3} to 10^{-6} cm, to describe powders whose mean diameter may range for ten micrometers to 10 nanometers. However, one should keep in mind that the grinding does not occur in vacuum, but most commonly either under argon or in ethanol. In the first case, powders have a tendency to stick to the grinding media, and in both cases, the velocity of the ball is strongly affected by aerodynamic or hydrodynamic interactions. Their movement can therefore not be predicted by a purely geometrical description.

C. Variation of the $\frac{\omega}{\Omega}$ ratio

In this last part of our analysis, we will only consider milling balls with a 1.0 cm diameter, as this is the condition we mostly use. However, I believe enough material is given here to reproduce quite rapidly the results, and am open to discussion if some consideration has to be taken with slightly different conditions.

In order to perform these calculations, the ball diameter is thus fixed to 1.0, in cell H1, and we are going to change only the value of ω, in cell B3, previously set at -2.

Figures 10 and 11 represent the movement of the ball, in the Cartesian coordinate system and in the rotating coordinate system attached to the rotation of the jar, respectively.

From Figure 10, it would seem that decreasing the $\frac{\omega}{\Omega}$ value has a similar effect than increasing the ball diameter, without changing the shock number nor the shock "energy", that is. On Figure 11, one can also see that, according to what was previously obtained in Figure 9, the shock angle also increases.



FIG. 10. Ball trajectory during the first full revolution of the jar in the Cartesian coordinate system. In blue is represented the trajectory of the center of the jar; in red, from bright to dark, are represented the trajectory of a ball with an $\frac{\omega}{\Omega}$ value of -2, -1.9, -1.8, -1.7 and -1.6.



FIG. 11. Ball trajectory during the first full revolution of the jar in the coordinate system associated with the rotating jar. In red, from bright to dark, are represented the trajectory of a ball with an $\frac{\omega}{\Omega}$ value of -2, -1.9, -1.8, -1.7 and -1.6.

It should be mentioned here that people using planetary ball mills allowing a separated control of Ω and ω

ω/Ω ratio	Maximum velocity $(m \cdot s^{-1} \cdot RPM^{-1})$	Max. relative velocity to the jar $(m \cdot s^{-1} \cdot RPM^{-1})$	Impact angle (°)	shock "frequency" (RPM^{-1})	Shock "energy" (see text)
-1.6	$8.07 \cdot 10^{-3}$	$4.486 \cdot 10^{-3}$	314.31 = -45.69	5.663	$1.139 \cdot 10^{-4}$
-1.7	$8.18 \cdot 10^{-3}$	$5.124 \cdot 10^{-3}$	322.09 = -37.91	4.952	$1.300 \cdot 10^{-4}$
-1.8	$8.31 \cdot 10^{-3}$	$5.736 \cdot 10^{-3}$	330.11 = -29.89	4.443	$1.461 \cdot 10^{-4}$
-1.9	$8.46 \cdot 10^{-3}$	$6.350 \cdot 10^{-3}$	339.13 = -20.87	4.046	$1.631 \cdot 10^{-4}$
-2.0	$8.65 \cdot 10^{-3}$	$6.916 \cdot 10^{-3}$	348.64 = -11.46	3.736	$1.786 \cdot 10^{-4}$
-2.1	$8.86 \cdot 10^{-3}$	$7.438 \cdot 10^{-3}$	357.83 = -2.16	3.497	$1.934 \cdot 10^{-4}$
-2.2	$9.12 \cdot 10^{-3}$	$7.931 \cdot 10^{-3}$	15.22	3.301	$2.075 \cdot 10^{-4}$
-2.3	$9.69 \cdot 10^{-3}$	$8.352 \cdot 10^{-3}$	17.56	3.158	$2.203 \cdot 10^{-4}$
-2.4	$10.96 \cdot 10^{-3}$	$8.682 \cdot 10^{-3}$	28.88	3.043	$2.294 \cdot 10^{-4}$

TABLE II. Milling parameters with different ω/Ω ratios.



FIG. 12. Ball trajectory during the first full revolution of the jar in the Cartesian coordinate system. In blue is represented the trajectory of the center of the jar; in red, from bright to dark, are represented the trajectory of a ball with an $\frac{\omega}{\Omega}$ value of -2, -2.1, -2.2, -2.3 and -2.4.



FIG. 13. Ball trajectory during the first full revolution of the jar in the coordinate system associated with the rotating jar. In red, from bright to dark, are represented the trajectory of a ball with an $\frac{\omega}{\Omega}$ value of -2, -2.1, -2.2, -2.3 and -2.4.

often have a different way of noting these velocities: they first take the absolute value for both, and instead of using ω, they refer to ω' (Eq.1). With such a notation, milling conditions such as $\omega/\Omega = 350/250$ would refer to an effective $\frac{\omega}{\Omega} = -1.7$.

Although I am not aware of any experiment using $\frac{\omega}{\Omega}$ values smaller than -2, these conditions are theoretically possible, and are represented on Figure 12 and 13.

Compiled results on the variation of the $\frac{\omega}{\Omega}$ value are represented in Table II.

In the considered ω/Ω ratios, we notice first a strong variation of the impact angle, ranging from -45 to 30°, thus allowing us to chose a grinding mode with a strong shear action; due to the ball diameter, this impact angle is significantly different from 0, and to get near this value, a ratio slightly under -2.1 should be chosen. Both velocities steadily increase when the ω/Ω ratio decrease. Despite the fact that the shock frequency decreases, the shock "energy" also increases, but is now always in the same order of magnitude.

VI. CONCLUSION

Using simple geometrical considerations, we defined a criteria to know whether a ball present in a milling jar has a free movement or is accelerated by the double rotation of the jar, around the solar center and around itself. From that point, we were able to describe the general movement of the ball, calculate its velocity, and determine the impact conditions. It is shown that the considered planetary ball mill present frontal impacts when the smallest balls are used; using greater sized balls, the optimal ω/Ω ratio decreases and is close to -2.1 for 1cm balls.